



## ARCHIVED - Archiving Content

### Archived Content

Information identified as archived is provided for reference, research or recordkeeping purposes. It is not subject to the Government of Canada Web Standards and has not been altered or updated since it was archived. Please contact us to request a format other than those available.

## ARCHIVÉE - Contenu archivé

### Contenu archivé

L'information dont il est indiqué qu'elle est archivée est fournie à des fins de référence, de recherche ou de tenue de documents. Elle n'est pas assujettie aux normes Web du gouvernement du Canada et elle n'a pas été modifiée ou mise à jour depuis son archivage. Pour obtenir cette information dans un autre format, veuillez communiquer avec nous.

This document is archival in nature and is intended for those who wish to consult archival documents made available from the collection of Public Safety Canada.

Some of these documents are available in only one official language. Translation, to be provided by Public Safety Canada, is available upon request.

Le présent document a une valeur archivistique et fait partie des documents d'archives rendus disponibles par Sécurité publique Canada à ceux qui souhaitent consulter ces documents issus de sa collection.

Certains de ces documents ne sont disponibles que dans une langue officielle. Sécurité publique Canada fournira une traduction sur demande.



Emergency Preparedness  
Canada

Protection civile  
Canada

An Interface Program for the  
Canada/United States  
Transborder Agreements Database

Technical Report 93-1

HV  
551.2  
I576  
1993

Canada

ID 1949 B

025.04 PRO

PSEPC Library/Bibliothèque de la SPPCC

Location: Main

September 2, 2004

ELLS-0681  
LIBRARY / BIBLIOTHÈQUE  
PSEPC/SPPCC  
OTTAWA (ONTARIO) K1A 0P3

**An Interface Program for the  
Canada/United States  
Transborder Agreements Database**

**Technical Report 93-1**

**by M. Provencher  
Evaluation and Analysis**

**Emergency Preparedness Canada**

**June 1993**

The contents of this document are the responsibility of the author, and do not necessarily reflect the opinion of Emergency Preparedness Canada.

Copyright of this document does not belong to the Crown.  
Proper authorization must be obtained from the author for  
any intended use.  
Les droits d'auteur du présent document n'appartiennent  
pas à l'État. Toute utilisation du contenu du présent  
document doit être approuvée préalablement par l'auteur.

## **Foreword**

This report aims to ease user access to the Transborder Agreements Database and is intended for people with some computer knowledge.

# Contents

Foreword	ii
Introduction	1
The Database and the Interface Program	2
How to Use the Program	5
Conclusion	9
Bibliography	10
Annex (program listings)	A-1

# An Interface Program for the Canada/United States Transborder Agreements Database

## Introduction

Under Contract No. 7A-872 from Emergency Preparedness Canada, a review of existing United States/Canada agreements related to emergency preparedness and emergency management cooperation was conducted. The results were presented in a Final Report dated 31 May 1988 [1] by the contractors Foottit-Mitchell and Associates.

Agreements were classified as Federal, Provincial/State, Local and Multi-Party. The Federal agreements were further classified by the department of prime interest.

Agreements were classified by the type and scale of emergency they were designed to meet, and by the type of agreed responsive action.

A manual (card) index of all agreements was produced. In all, 157 agreements of one type or another were identified as relevant to the subject of emergency preparedness and emergency management.

Those agreements were incorporated by Dr. G.W. Frank into a computer database using the Relational Database Management System (RDBMS) FoxBASE on the Macintosh computer. His work was documented in Technical Report 92-2 [2].

The database has since been slightly modified to allow for easy searches of agreements using the 56 keywords listed in [1]. It now contains 160 agreements.

This document describes an interface program which was created to facilitate access to the database in various modes of operation; for example, searching for agreements, modifying them, adding new ones and printing them.

## The Database and the Interface Program

The database currently consists of two files: "TBA.dbf," which contains most of the details (see p. 3, Figure 1) and "keywords.dbf" which, as its name indicates, contains all the keywords that may be used to describe emergency events (see p. 4, Figure 2). Both files are linked through the agreement identifier agr\_id. Each of the 160 entries in the files TBA.dbf and keywords.dbf represents information on a single agreement i.e. its title, content, the states and provinces involved, the U.S. and Canadian levels, applicable keywords, etc.

Many fields are required in order to define/identify an agreement within the database. To be specific, 103 fields are used. With the standard FoxBASE display of information, as Figure 3 (p. 5) illustrates, the task of updating the database is very cumbersome due to the need to scroll the screens for all the information corresponding to a single entry. This was the reason behind the creation of the interface program "TBAtool."

TBAtool is in fact the main program of a series of other programs all designed to make access to the Transborder Agreement (TBA) database easier. All programs are FoxBASE programs and are listed in the annex.

Along with the programs, a memory file "tool.mem" needs to be accessed. This memory file contains four arrays used to generate appropriate information from the pop-up menus. They are:

uil (U.S. involvement level) with values: " ", "LOCAL", "STATE",  
"FED", "MTL"  
cil (Canadian involvement level) with values: " ", "LOCAL",  
"PROV", "FED", "MTL"  
ste (scale of the emergency) with values: " ", "LOC", "REG",  
"NAT"  
opc (authorization for) with values: " ", "OPER", "PLAN"



**Figure 1: Structure of TBA.dbf file**

Field	Field Name	Type	Width	Dec
1	agr_id	Numeric	5	
2	title	Character	100	
3	date_in	Character	10	
4	date_rev	Character	10	
5	date_exp	Character	10	
6	us_auth	Character	10	
7	can_auth	Character	10	
8	content	Memo	10	
9	ak	Logical	1	
10	wa	Logical	1	
11	id	Logical	1	
12	mt	Logical	1	
13	nd	Logical	1	
14	mn	Logical	1	
15	wi	Logical	1	
16	mi	Logical	1	
17	oh	Logical	1	
18	ny	Logical	1	
19	vt	Logical	1	
20	nh	Logical	1	
21	me	Logical	1	
22	or	Logical	1	
23	ca	Logical	1	
24	yt	Logical	1	
25	bc	Logical	1	
26	ab	Logical	1	
27	sk	Logical	1	
28	mb	Logical	1	
29	on	Logical	1	
30	pq	Logical	1	
31	nb	Logical	1	
32	us_ref	Character	8	
33	can_ref	Character	8	
34	us_lev	Character	5	
35	cdn_lev	Character	5	
36	e_type1	Character	6	
37	e_type2	Character	6	
38	e_type3	Character	6	
39	scale	Character	3	
40	r_type1	Character	4	
41	r_type2	Character	4	
42	r_type3	Character	4	
43	oper	Character	4	
44	comment	Memo	10	
45	reference	Memo	10	
46	impacts_on	Memo	10	
** Total **			282	

**Figure 2: Structure of Keywords.dbf file**

Structure for database:		Internal 40:Can/US TBA db:keywords.dbf		
Number of data records:		160		
Date of last update:		06/03/92		
Field	Field Name	Type	Width	Dec
1	agr_id	Numeric	5	
2	air_pollut	Logical	1	
3	air_qualit	Logical	1	
4	air_traffi	Logical	1	
5	ambulance	Logical	1	
6	atomic_ene	Logical	1	
7	audit	Logical	1	
8	bomb_threa	Logical	1	
9	broadcast	Logical	1	
10	coastal_wa	Logical	1	
11	communicat	Logical	1	
12	consultati	Logical	1	
13	cooperatio	Logical	1	
14	data_excha	Logical	1	
15	earthquake	Logical	1	
16	emerg_plan	Logical	1	
17	federal_co	Logical	1	
18	fire_fight	Logical	1	
19	floods	Logical	1	
20	forest_fir	Logical	1	
21	frequencie	Logical	1	
22	hazard_was	Logical	1	
23	ice_breaki	Logical	1	
24	industry	Logical	1	
25	local_comp	Logical	1	
26	mail	Logical	1	
27	marin_traf	Logical	1	
28	milit_coop	Logical	1	
29	mobilizati	Logical	1	
30	mountains	Logical	1	
31	navigation	Logical	1	
32	nuclear_re	Logical	1	
33	oil_pollut	Logical	1	
34	planning	Logical	1	
35	power	Logical	1	
36	priorities	Logical	1	
37	prov_compr	Logical	1	
38	rad_report	Logical	1	
39	radio	Logical	1	
40	refugee_co	Logical	1	
41	r_and_d	Logical	1	
42	safety	Logical	1	
43	satellites	Logical	1	
44	sar	Logical	1	
45	security	Logical	1	
46	state_comp	Logical	1	
47	strikes	Logical	1	
48	testing	Logical	1	
49	trade	Logical	1	
50	training	Logical	1	
51	transporta	Logical	1	
52	vessel_acc	Logical	1	
53	volcano	Logical	1	
54	water_cont	Logical	1	
55	water_poll	Logical	1	
56	water_qual	Logical	1	
57	weather	Logical	1	
** Total **	62			

Figure 3: Standard FoxBASE table views

Date_in	Date_rev	Date_exp	Us_auth	Can_auth	Content	Ak
08/11/1972	24/08/1982		SEC STATE	AMBASS	memo	
20/08/1971	14/09/1983		FED COMM	DEPT COMM	memo	
06/05/1975	29/12/1978		AMBASS	SEC STATE	memo	
01/01/1976			FED COMM	DEPT COMM	memo	
04/06/1973			FED COMM	DEPT COMM	memo	
15/12/1976	18/11/1983		FED COMM	DEPT COMM	memo	
13/01/1977	24/12/1981		FED COMM	DEPT COMM	memo	
01/01/1979			SEC COMM	DEPT COMM	memo	
01/01/1981			FED COMM	DEPT COMM	memo	

Keywords							
Agr_id	Air_pollut	Air_qualit	Air_traffi	Ambulance	Atomic_ene	Audit	Bo
13							
14							
15							
16							
17							
18							
19							

## How to Use the Program

To be able to run the program TBAtool, you have to verify that the program files, database files, memory file, index files "agr.idx" and "agr2.idx," and report file "repcaus.frx" are in the same folder.<sup>1</sup> Then you double-click on TBAtool.prg and it starts! Alternatively, if the RDBMS FoxBASE is already active, you go under the menu item **Program**, select **Do...** and

<sup>1</sup> The files agr.idx, agr2.idx, keywords.dbf, repcaus.frx, repcaus.vue, TBA.dbf, TBA.dbt, tool.mem in addition to the ones referred to on page A-1 need to be in the same folder.

(proceeding through directories/folders as necessary) the program TBAtool.prg, which will start upon selection.


In designing the program it was decided that the standard mode would be one of only displaying the information, thus preventing the user from accidentally modifying the information already recorded. TBAtool will show the first agreement in the same format as agreement number 11 presented on Figure 4. All the information (every field) on the agreement is displayed there except for the keywords which are available in editing mode.

Figure 4: Display of an agreement

File Edit Position Record Utilities			
<b>Title</b> EXPERIMENTAL SATELLITE COMMUNICATIONS		<b>Agreement ID</b> 11 <b>Canadian Reference Block</b> CF-DC-08 <b>US Reference Block</b> UF-NA-01	
<b>Content</b> EXCHANGE OF NOTES GIVING AGREEMENTS BETWEEN CANADA AND THE UNITED STATES ON COOPERATION IN EXPERIMENTAL SATELLITE COMMUNICATIONS:		<b>Initial Date</b> 23/08/1963 <b>Revision Date</b> 27/04/1971 <b>Expiry Date</b> / /	
<b>Canadian Authority</b> DEPT COMM <b>US Authority</b> NASA		<b>Emergency Type Addressed</b> WAR-04 <b>Response Type Addressed</b> R-24	
<b>States involved</b> <ul style="list-style-type: none"> <li><input type="checkbox"/> Alaska</li> <li><input type="checkbox"/> Washington</li> <li><input type="checkbox"/> Idaho</li> <li><input type="checkbox"/> Montana</li> <li><input type="checkbox"/> North Dakota</li> <li><input type="checkbox"/> Minnesota</li> <li><input type="checkbox"/> Wisconsin</li> <li><input type="checkbox"/> Michigan</li> <li><input type="checkbox"/> Ohio</li> <li><input type="checkbox"/> New York</li> <li><input type="checkbox"/> Vermont</li> <li><input type="checkbox"/> New Hampshire</li> <li><input type="checkbox"/> Maine</li> <li><input type="checkbox"/> Oregon</li> <li><input type="checkbox"/> California</li> </ul>		<b>US Involvement</b> <input type="text" value="FED"/> <b>Canadian Involvement</b> <input type="text" value="FED"/> <b>Scale of Emergency</b> <input type="text" value="NAT"/> <b>Authorization is for</b> <input type="text" value="OPER"/>	
		<b>Impact on other Agencies</b> US FEDERAL COMMUNICATIONS COMMISSION	
		<b>References</b> 14 UST 1701 TIAS 474 22 UST 713 TIAS 713 CTS 1967 NO.13 <b>Comments</b>	
		<b>Provinces/Territories involved</b> <ul style="list-style-type: none"> <li><input type="checkbox"/> Yukon</li> <li><input type="checkbox"/> British Columbia</li> <li><input type="checkbox"/> Alberta</li> <li><input type="checkbox"/> Saskatchewan</li> <li><input type="checkbox"/> Manitoba</li> <li><input type="checkbox"/> Ontario</li> <li><input type="checkbox"/> Quebec</li> <li><input type="checkbox"/> New Brunswick</li> </ul>	


Notice the menu items "Position," "Record" and "Utilities." "Position" (Figure 5) allows the user to navigate through the records in a simple way. The user can go to the top or bottom of the file, and shift records to the next or to the previous, always displaying the information on the current record. This positioning can be done through the menu or alternatively through the keys when applicable (e.g. Command-N, Command-P).

Figure 5: The Position Menu



However, when one wants to see (a) specific record(s) the menu item "Record" has to be used (Figure 6). The "Locate" option calls the FoxBASE Expression Builder dialog. This is where the user has to determine the set of records wanted. A record is included in the set if its attributes satisfy the prescribed logical statement used in the Expression Builder. If a matching record is found, the Continue option is enabled so that the capability is there to search for any more records. If a matching record is not found, the ALERT "No more matching records!" is displayed.

Figure 6: The Record Menu



The other options under the Record menu have to be used carefully since they will modify the content of the database. Thus, if one only wants to see the keywords in the editing mode, one has to make sure one does not accidentally hit a key. The Edit option, as well as the Append option, are very similar in their presentation of the record (see p. 8, Figure 7).

Figure 7: Agreement in Edit mode

File Edit Position Record Utilities

Title  
EXPERIMENTAL SATELLITE COMMUNICATIONS

Agreement ID 11  
Canadian Reference Block CF-DC-08  
US Reference Block UF-NA-01

Content  
EXCHANGE OF NOTES GIVING AGREEMENTS BETWEEN CANADA AND THE UNITED STATES ON COOPERATION IN EXPERIMENTAL SATELLITE COMMUNICATIONS:

Initial Date 23/08/1963  
Revision Date 27/04/1971  
Expiry Date / /

Canadian Authority DEPT COMM  
US Authority NASA

Emergency Type Addressed WAR-04  
Response Type Addressed R-24

States involved

- Alaska
- Washington
- Idaho
- Montana
- North Dakota
- Minnesota
- Wisconsin
- Michigan
- Ohio
- New York
- Vermont
- New Hampshire
- Maine
- Oregon
- California

US Involvement FED  
Canadian Involvement FED  
Scale of Emergency NAT  
Authorization is for OPER

Provinces/Territories involved

- Yukon
- British Columbia
- Alberta
- Saskatchewan
- Manitoba
- Ontario
- Quebec
- New Brunswick

Impact on other Agencies  
US FEDERAL COMMUNICATIONS COMMISSION

References  
14 UST 1701  
TIAS 474  
22 UST 713  
TIAS 713  
CTS 1967 NO.13

Comments

Finished Keywords

Figure 8: Keywords

File Edit Position Record Utilities

**Keywords**

- Air Pollution
- Air Quality
- Air Traffic
- Ambulance
- Atomic Energy
- Audit
- Bomb Threat
- Broadcast
- Coastal Waters
- Communications
- Consultation
- Cooperation
- Data Exchange
- Earthquakes
- Emergency Planning
- Federal Comprehensive
- Fire Fighting
- Floods
- Forest Fires
- Frequencies
- Hazardous Waste
- Ice Breaking
- Industry
- Local Comprehensive
- Mail
- Marine Traffic
- Military Cooperation
- Mobilization
- Mountains
- Navigation
- Nuclear Research
- Oil Pollution
- Planning
- Power
- Priorities
- Provincial Comprehensive
- Radiation Reporting
- Radio
- Refuge Control
- Research and Development
- Safety
- Satellites
- Search and Rescue
- Security
- State Comprehensive
- Strikes
- Testing
- Trade
- Training
- Transportation
- Vessel Accident
- Volcano
- Water Control
- Water Pollution
- Water Quality
- Weather


Finished Back

The principal difference is that in the Append mode there is nothing in the fields except for the Agreement ID, which is automatically generated. In either case the agreement id cannot be changed. Please note the "Finished" and "Keywords" buttons at the bottom of the screen. "Finished" allows the user to get out of Edit/Append mode at any time. "Keywords" allows the user access to the keywords corresponding to this record (see p. 8, Figure 8). Notice that there again there is the option of getting out of Edit/Append mode or going back to the previous screen.

The Delete option (Figure 6) is straight forward. It just deletes the current record.

The Utilities menu (Figure 9) has two basic options. The "Report" option allows the user to print the current agreement (the one on the screen). "Quit Application" may be used when all the desired operations/activities on the database are completed.

Figure 9: The Utilities Menu



## Conclusion

The purpose of this paper was to explain an interface program recently created to facilitate access to the Transborder Agreements Database. Illustrative screens are presented to assist the users. Full program listings are given so as to allow future adjustments.

## **Bibliography**

1. *A Review of United States/Canada Agreements related to Emergency Preparedness/Management* by Footitt-Mitchell and Associates, 31 May 1988.
2. *Transborder Agreements Computer Database on Emergency Preparedness/Management* by Dr. G.W. Frank, EPC Technical Report 92-2, November 1992.



## Annex

### Program Listings

#### Programs

Add\_Rec.prg.....A-2  
Edit\_Rec.prg.....A-3  
Err\_Proc.prg.....A-4  
Find\_Rec.prg.....A-5  
GETs.prg.....A-6  
GETs1.prg.....A-8  
GETs2.prg.....A-9  
Initialize.prg.....A-10  
MenuProc.prg.....A-11  
SAYs.prg.....A-14  
SAYs2.prg.....A-15  
TBAtool.prg.....A-16  
Update.prg.....A-18  
UpdtMenu.prg.....A-19

```

* Program...: Add_Rec
*-----
* ---Program to add a record
*-----
* ---Disable all user-defined menus in the menu bar.
MENU OFF 1
MENU OFF 2
MENU OFF 3
*
SELECT A
GOTO BOTTOM
STORE Tba->Agr_id TO numero
APPEND BLANK
REPLACE Tba->Agr_id WITH numero+1
SELECT B
APPEND BLANK
REPLACE Keywords->Agr_id WITH numero+1
STORE 0 TO page2
STORE 1 TO page1
STORE 0 TO finished
DO WHILE finished=0
  CLEAR
  DO SAYS
  DO WHILE (page2=0 .AND. finished=0)
    @ PIXELS 21,459 GET Tba->Agr_id STYLE 0 FONT "Times",14 SIZE 15,52 COLOR 0,0,0,-1,-1,-1
    CLEAR GETS
    DO GETS
    DO GETs1
    @ PIXELS 420,550 GET page2 STYLE 0 FONT "Chicago",10 PICTURE "@*V Keywords" SIZE 2068,63 COLOR 0,0,0,-1,-1,-1
    @ PIXELS 420,470 GET finished STYLE 0 FONT "Chicago",10 PICTURE "@*V Finished" SIZE 2068,63 COLOR 0,0,0,-1,-1,-1
    READ
    REPLACE us_lev WITH UIL(temp1)
    REPLACE cdn_lev WITH CIL(temp2)
    REPLACE scale WITH STE(temp3)
    REPLACE oper WITH OPC(temp4)
    DO Update
  ENDDO
  STORE 0 TO page1
  CLEAR
  DO SAYS2
  DO WHILE (page1=0 .AND. finished=0)
    DO GETS2
    @ PIXELS 310,441 GET page1 STYLE 0 FONT "Times",4364 PICTURE "@*V Back" SIZE 2068,63 COLOR 0,0,0,-1,-1,-1
    @ PIXELS 310,361 GET finished STYLE 0 FONT "Chicago",10 PICTURE "@*V Finished" SIZE 2068,63 COLOR 0,0,0,-1,-1,-1
    READ
  ENDDO
  STORE 0 TO page2
ENDDO
SCREEN 1 TOP
* ---Enable menu system again, now that record is added
MENU ON 1
MENU ON 2
MENU ON 3
RETURN
* EOF : Add_Rec.prg

```

```

* Program...: Edit_Rec
*-----
* ---Program to edit a record
*-----
* ---Disable all user-defined menus in the menu bar.
*      This prevents user from making another selection while in edit mode.
MENU OFF 1
MENU OFF 2
MENU OFF 3
*
STORE 0 TO page2
STORE 1 TO page1
STORE 0 TO finished
DO WHILE finished=0
  CLEAR
  DO SAYS
  DO WHILE (page2=0 .AND. finished=0)
    @ PIXELS 21,459 GET Tba->Agr_id STYLE 0 FONT "Times",14 SIZE 15,52 COLOR 0,0,0,-1,-1,-1
    CLEAR GETS
    DO GETS
    DO GETs1
    @ PIXELS 420,550 GET page2 STYLE 0 FONT "Chicago",10 PICTURE "@*V Keywords" SIZE 2068,63 COLOR 0,0,0,-1,-1,-1
    @ PIXELS 420,470 GET finished STYLE 0 FONT "Chicago",10 PICTURE "@*V Finished" SIZE 2068,63 COLOR 0,0,0,-1,-1,-1
    READ
    REPLACE us_lev WITH UIL(temp1)
    REPLACE cdn_lev WITH CIL(temp2)
    REPLACE scale WITH STE(temp3)
    REPLACE oper WITH OPC(temp4)
    DO Update
  ENDDO
  STORE 0 TO page1
  CLEAR
  DO SAYS2
  DO WHILE (page1=0 .AND. finished=0)
    DO GETS2
    @ PIXELS 310,441 GET page1 STYLE 0 FONT "Times",4364 PICTURE "@*V Back" SIZE 2068,63 COLOR 0,0,0,-1,-1,-1
    @ PIXELS 310,361 GET finished STYLE 0 FONT "Chicago",10 PICTURE "@*V Finished" SIZE 2068,63 COLOR 0,0,0,-1,-1,-1
    READ
  ENDDO
  STORE 0 TO page2
ENDDO
SCREEN 1 TOP
* ---Enable menu system again, since editing is complete
MENU ON 1
MENU ON 2
MENU ON 3
RETURN
* EOF : Edit_Rec.prg

```

\* Program...: Err\_Proc

\*-----

\* ---Program to handle errors

\*-----

\* ---Display an alert that contains the error message

ALERT STOP 2 MESSAGE()

RETURN

\*EOF : Err\_Proc.prg

```

* Program...: Find_Rec
*-----
* ---Program to find a record
*-----
* ---Store current record number to lastrec variable
STORE RECNO() TO lastrec
* ---Display the Expression Builder dialog with the message LOCATE FOR.
*   Store user input to lookkey variable.
GETEXPR "LOCATE FOR:" TO lookkey
* ---If the expression builder returned an expression
IF .NOT. (lookkey = "")
  * ---LOCATE for user-specified lookkey, note use of macro substitution.
  LOCATE FOR &lookkey
  * ---If a matching record was found
  IF FOUND()
    * ---Update the menu system (enables the CONTINUE option)
    DO UpdtMenu
  ELSE && a matching record was not found
    * ---Display a message -CHR(13) causes line break-
    ALERT NOTE 1 "No record found for:",CHR(13),lookkey
    * ---Return to previous record
    GOTO lastrec
  ENDIF
ENDIF
RETURN
* EOF : Find_Rec.prg

```

```

* Program...: GETS
*-----
* ---Input Program
*-----
*
SELECT A
@ PIXELS 25,7 GET Tba->Title STYLE 0 FONT "Times",14 SIZE 45,353 COLOR 0,0,0,-1,-1,-1
@ PIXELS 38,531 GET Tba->can_ref STYLE 0 FONT "Times",14 PICTURE "AA-AA-99" SIZE 15,80 COLOR 0,0,0,-1,-1,-1
@ PIXELS 56,505 GET Tba->us_ref STYLE 0 FONT "Times",14 PICTURE "AA-AA-99" SIZE 15,83 COLOR 0,0,0,-1,-1,-1
@ PIXELS 84,7 GET Tba->content STYLE 0 FONT "Times",14 SCROLL SIZE 60,425 COLOR 0,0,0,-1,-1,-1
@ PIXELS 86,518 GET Tba->Date_in STYLE 0 FONT "Times",14 PICTURE "##/##/####" SIZE 15,79 COLOR 0,0,0,-1,-1,-1
@ PIXELS 105,537 GET Tba->Date_rev STYLE 0 FONT "Times",14 PICTURE "##/##/####" SIZE 15,70 COLOR 0,0,0,-1,-1,-1
@ PIXELS 124,522 GET Tba->Date_exp STYLE 0 FONT "Times",14 PICTURE "##/##/####" SIZE 15,79 COLOR 0,0,0,-1,-1,-1
@ PIXELS 158,123 GET Tba->can_auth STYLE 0 FONT "Times",14 SIZE 15,97 COLOR 0,0,0,-1,-1,-1
@ PIXELS 176,123 GET Tba->us_auth STYLE 0 FONT "Times",14 SIZE 15,97 COLOR 0,0,0,-1,-1,-1
@ PIXELS 152,410 GET Tba->e_type1 STYLE 0 FONT "Times",14 SIZE 15,52 COLOR 0,0,0,-1,-1,-1
@ PIXELS 152,482 GET Tba->e_type2 STYLE 0 FONT "Times",14 SIZE 15,52 COLOR 0,0,0,-1,-1,-1
@ PIXELS 152,554 GET Tba->e_type3 STYLE 0 FONT "Times",14 SIZE 15,52 COLOR 0,0,0,-1,-1,-1
@ PIXELS 172,410 GET Tba->r_type1 STYLE 0 FONT "Times",14 SIZE 15,52 COLOR 0,0,0,-1,-1,-1
@ PIXELS 172,482 GET Tba->r_type2 STYLE 0 FONT "Times",14 SIZE 15,52 COLOR 0,0,0,-1,-1,-1
@ PIXELS 172,554 GET Tba->r_type3 STYLE 0 FONT "Times",14 SIZE 15,52 COLOR 0,0,0,-1,-1,-1
DO CASE
  CASE SUBSTR(Tba->us_lev,1,1)=" "
    temp1=1
  CASE SUBSTR(Tba->us_lev,1,1)="L"
    temp1=2
  CASE SUBSTR(Tba->us_lev,1,1)="S"
    temp1=3
  CASE SUBSTR(Tba->us_lev,1,1)="F"
    temp1=4
  CASE SUBSTR(Tba->us_lev,1,1)="M"
    temp1=5
ENDCASE
@ PIXELS 202,298 GET temp1 STYLE 0 FONT "Chicago",12 PICTURE "@^ NONE;LOCAL;STATE;FED;MTL" SIZE 19,57 COLOR 0,0,0,-1,-1,-1
DO CASE
  CASE SUBSTR(Tba->cdn_lev,1,1)=" "
    temp2=1
  CASE SUBSTR(Tba->cdn_lev,1,1)="L"
    temp2=2
  CASE SUBSTR(Tba->cdn_lev,1,1)="P"
    temp2=3
  CASE SUBSTR(Tba->cdn_lev,1,1)="F"
    temp2=4
  CASE SUBSTR(Tba->cdn_lev,1,1)="M"
    temp2=5
ENDCASE
@ PIXELS 226,298 GET temp2 STYLE 0 FONT "Chicago",12 PICTURE "@^ NONE;LOCAL;PROV;FED;MTL" SIZE 19,57 COLOR 0,0,0,-1,-1,-1
DO CASE
  CASE SUBSTR(Tba->scale,1,1)=" "
    temp3=1
  CASE SUBSTR(Tba->scale,1,1)="L"
    temp3=2
  CASE SUBSTR(Tba->scale,1,1)="R"
    temp3=3
  CASE SUBSTR(Tba->scale,1,1)="N"

```

```
temp3=4
ENDCASE
@ PIXELS 250,298 GET temp3 STYLE 0 FONT "Chicago",12 PICTURE "@^ NONE;LOC;REG;NAT" SIZE 19,57 COLOR 0,0,0,-1,-1,-1
DO CASE
CASE SUBSTR(Tba->oper,1,1)=" "
temp4=1
CASE SUBSTR(Tba->oper,1,1)="O"
temp4=2
CASE SUBSTR(Tba->oper,1,1)="P"
temp4=3
ENDCASE
@ PIXELS 273,298 GET temp4 STYLE 0 FONT "Chicago",12 PICTURE "@^ NONE;OPER;PLAN" SIZE 19,57 COLOR 0,0,0,-1,-1,-1
@ PIXELS 210,383 GET Tba->impacts_on STYLE 0 FONT "Times",12 SCROLL SIZE 60,234 COLOR 0,0,0,-1,-1,-1
@ PIXELS 287,383 GET Tba->reference STYLE 0 FONT "Times",12 SCROLL SIZE 60,234 COLOR 0,0,0,-1,-1,-1
@ PIXELS 364,383 GET Tba->comment STYLE 0 FONT "Times",12 SCROLL SIZE 60,234 COLOR 0,0,0,-1,-1,-1
*
RETURN
*
* EOF : GETs.prg
```

\* Program...: GETs1

\*-----

\* ---Input Program

\*-----

\*

SELECT A

DO Initialize

@ PIXELS 215,14 GET state(1) STYLE 65536 FONT "Times",268 PICTURE "@\*C Alaska" SIZE 13,60 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 230,14 GET state(2) STYLE 65536 FONT "Times",268 PICTURE "@\*C Washington" SIZE 13,89 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 245,14 GET state(3) STYLE 65536 FONT "Times",268 PICTURE "@\*C Idaho" SIZE 13,54 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 260,14 GET state(4) STYLE 65536 FONT "Times",268 PICTURE "@\*C Montana" SIZE 13,70 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 275,14 GET state(5) STYLE 65536 FONT "Times",268 PICTURE "@\*C North Dakota" SIZE 13,98 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 290,14 GET state(6) STYLE 65536 FONT "Times",268 PICTURE "@\*C Minnesota" SIZE 13,80 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 305,14 GET state(7) STYLE 65536 FONT "Times",268 PICTURE "@\*C Wisconsin" SIZE 13,81 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 320,14 GET state(8) STYLE 65536 FONT "Times",268 PICTURE "@\*C Michigan" SIZE 13,74 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 335,14 GET state(9) STYLE 65536 FONT "Times",268 PICTURE "@\*C Ohio" SIZE 13,49 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 350,14 GET state(10) STYLE 65536 FONT "Times",268 PICTURE "@\*C New York" SIZE 13,80 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 365,14 GET state(11) STYLE 65536 FONT "Times",268 PICTURE "@\*C Vermont" SIZE 13,70 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 380,14 GET state(12) STYLE 65536 FONT "Times",268 PICTURE "@\*C New Hampshire" SIZE 13,113 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 395,14 GET state(13) STYLE 65536 FONT "Times",268 PICTURE "@\*C Maine" SIZE 13,56 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 410,14 GET state(14) STYLE 65536 FONT "Times",268 PICTURE "@\*C Oregon" SIZE 13,63 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 425,14 GET state(15) STYLE 65536 FONT "Times",268 PICTURE "@\*C California" SIZE 13,79 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 319,159 GET prov(1) STYLE 65536 FONT "Times",268 PICTURE "@\*C Yukon" SIZE 13,59 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 334,159 GET prov(2) STYLE 65536 FONT "Times",268 PICTURE "@\*C British Columbia" SIZE 13,119 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 349,159 GET prov(3) STYLE 65536 FONT "Times",268 PICTURE "@\*C Alberta" SIZE 13,63 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 364,159 GET prov(4) STYLE 65536 FONT "Times",268 PICTURE "@\*C Saskatchewan" SIZE 13,101 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 379,159 GET prov(5) STYLE 65536 FONT "Times",268 PICTURE "@\*C Manitoba" SIZE 13,74 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 394,159 GET prov(6) STYLE 65536 FONT "Times",268 PICTURE "@\*C Ontario" SIZE 13,64 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 409,159 GET prov(7) STYLE 65536 FONT "Times",268 PICTURE "@\*C Quebec" SIZE 13,63 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 424,159 GET prov(8) STYLE 65536 FONT "Times",268 PICTURE "@\*C New Brunswick" SIZE 13,113 COLOR 0,0,0,-1,-1,-1

\*

RETURN

\*

\* EOF : GETs1.prg



\* Program...: GETs2

\*-----  
\* ---Input Program for keywords  
\*-----

```
SELECT B
@ PIXELS 40,9 GET Keywords->air_pollut STYLE 65536 FONT "Times",268 PICTURE "@*C Air Pollution" SIZE 13,96 COLOR 0,0,0,-1,-1,-1
@ PIXELS 55,9 GET Keywords->air_qualit STYLE 65536 FONT "Times",268 PICTURE "@*C Air Quality" SIZE 13,85 COLOR 0,0,0,-1,-1,-1
@ PIXELS 70,9 GET Keywords->air_traffi STYLE 65536 FONT "Times",268 PICTURE "@*C Air Traffic" SIZE 13,83 COLOR 0,0,0,-1,-1,-1
@ PIXELS 85,9 GET Keywords->ambulance STYLE 65536 FONT "Times",268 PICTURE "@*C Ambulance" SIZE 13,84 COLOR 0,0,0,-1,-1,-1
@ PIXELS 100,9 GET Keywords->atomic_ene STYLE 65536 FONT "Times",268 PICTURE "@*C Atomic Energy" SIZE 13,106 COLOR 0,0,0,-1,-1,-1
@ PIXELS 115,9 GET Keywords->audit STYLE 65536 FONT "Times",268 PICTURE "@*C Audit" SIZE 13,53 COLOR 0,0,0,-1,-1,-1
@ PIXELS 130,9 GET Keywords->bomb_threa STYLE 65536 FONT "Times",268 PICTURE "@*C Bomb Threat" SIZE 13,95 COLOR 0,0,0,-1,-1,-1
@ PIXELS 145,9 GET Keywords->broadcast STYLE 65536 FONT "Times",268 PICTURE "@*C Broadcast" SIZE 13,78 COLOR 0,0,0,-1,-1,-1
@ PIXELS 160,9 GET Keywords->coastal wa STYLE 65536 FONT "Times",268 PICTURE "@*C Coastal Waters" SIZE 13,107 COLOR 0,0,0,-1,-1,-1
@ PIXELS 175,9 GET Keywords->communicat STYLE 65536 FONT "Times",268 PICTURE "@*C Communications" SIZE 13,116 COLOR 0,0,0,-1,-1,-1
@ PIXELS 190,9 GET Keywords->consultati STYLE 65536 FONT "Times",268 PICTURE "@*C Consultation" SIZE 13,94 COLOR 0,0,0,-1,-1,-1
@ PIXELS 205,9 GET Keywords->cooperatio STYLE 65536 FONT "Times",268 PICTURE "@*C Cooperation" SIZE 13,91 COLOR 0,0,0,-1,-1,-1
@ PIXELS 220,9 GET Keywords->data_excha STYLE 65536 FONT "Times",268 PICTURE "@*C Data Exchange" SIZE 13,105 COLOR 0,0,0,-1,-1,-1
@ PIXELS 235,9 GET Keywords->earthquake STYLE 65536 FONT "Times",268 PICTURE "@*C Earthquake" SIZE 13,85 COLOR 0,0,0,-1,-1,-1
@ PIXELS 250,9 GET Keywords->emerg_plan STYLE 65536 FONT "Times",268 PICTURE "@*C Emergency Planning" SIZE 13,85 COLOR 0,0,0,-1,-1,-1
@ PIXELS 265,9 GET Keywords->federal_co STYLE 65536 FONT "Times",268 PICTURE "@*C Federal Comprehensive" SIZE 13,85 COLOR 0,0,0,-1,-1,-1
@ PIXELS 280,9 GET Keywords->fire fight STYLE 65536 FONT "Times",268 PICTURE "@*C Fire Fighting" SIZE 13,85 COLOR 0,0,0,-1,-1,-1
@ PIXELS 295,9 GET Keywords->floods STYLE 65536 FONT "Times",268 PICTURE "@*C Floods" SIZE 13,85 COLOR 0,0,0,-1,-1,-1
@ PIXELS 310,9 GET Keywords->forest fir STYLE 65536 FONT "Times",268 PICTURE "@*C Forest Fires" SIZE 13,85 COLOR 0,0,0,-1,-1,-1
@ PIXELS 40,162 GET Keywords->frequencie STYLE 65536 FONT "Times",268 PICTURE "@*C Frequencies" SIZE 13,90 COLOR 0,0,0,-1,-1,-1
@ PIXELS 55,162 GET Keywords->hazard was STYLE 65536 FONT "Times",268 PICTURE "@*C Hazardous Waste" SIZE 13,120 COLOR 0,0,0,-1,-1,-1
@ PIXELS 70,162 GET Keywords->ice breaki STYLE 65536 FONT "Times",268 PICTURE "@*C Ice Breaking" SIZE 13,94 COLOR 0,0,0,-1,-1,-1
@ PIXELS 85,162 GET Keywords->industry STYLE 65536 FONT "Times",268 PICTURE "@*C Industry" SIZE 13,70 COLOR 0,0,0,-1,-1,-1
@ PIXELS 100,162 GET Keywords->local comp STYLE 65536 FONT "Times",268 PICTURE "@*C Local Comprehensive" SIZE 13,144 COLOR 0,0,0,-1,-1,-1
@ PIXELS 115,162 GET Keywords->mail STYLE 65536 FONT "Times",268 PICTURE "@*C Mail" SIZE 13,47 COLOR 0,0,0,-1,-1,-1
@ PIXELS 130,162 GET Keywords->marin traf STYLE 65536 FONT "Times",268 PICTURE "@*C Marine Traffic" SIZE 13,104 COLOR 0,0,0,-1,-1,-1
@ PIXELS 145,162 GET Keywords->milit coop STYLE 65536 FONT "Times",268 PICTURE "@*C Military Cooperation" SIZE 13,140 COLOR 0,0,0,-1,-1,-1
@ PIXELS 160,162 GET Keywords->mobilizati STYLE 65536 FONT "Times",268 PICTURE "@*C Mobilization" SIZE 13,93 COLOR 0,0,0,-1,-1,-1
@ PIXELS 175,162 GET Keywords->mountains STYLE 65536 FONT "Times",268 PICTURE "@*C Mountains" SIZE 13,81 COLOR 0,0,0,-1,-1,-1
@ PIXELS 190,162 GET Keywords->navigation STYLE 65536 FONT "Times",268 PICTURE "@*C Navigation" SIZE 13,84 COLOR 0,0,0,-1,-1,-1
@ PIXELS 205,162 GET Keywords->nuclear_re STYLE 65536 FONT "Times",268 PICTURE "@*C Nuclear Research" SIZE 13,121 COLOR 0,0,0,-1,-1,-1
@ PIXELS 220,162 GET Keywords->oil_pollut STYLE 65536 FONT "Times",268 PICTURE "@*C Oil Pollution" SIZE 13,95 COLOR 0,0,0,-1,-1,-1
@ PIXELS 235,162 GET Keywords->planning STYLE 65536 FONT "Times",268 PICTURE "@*C Planning" SIZE 13,72 COLOR 0,0,0,-1,-1,-1
@ PIXELS 250,162 GET Keywords->power STYLE 65536 FONT "Times",268 PICTURE "@*C Power" SIZE 13,58 COLOR 0,0,0,-1,-1,-1
@ PIXELS 265,162 GET Keywords->priorities STYLE 65536 FONT "Times",268 PICTURE "@*C Priorities" SIZE 13,75 COLOR 0,0,0,-1,-1,-1
@ PIXELS 280,162 GET Keywords->prov compr STYLE 65536 FONT "Times",268 PICTURE "@*C Provincial Comprehensive" SIZE 13,171 COLOR 0,0,0,-1,-1,-1
@ PIXELS 295,162 GET Keywords->rad report STYLE 65536 FONT "Times",268 PICTURE "@*C Radiation Reporting" SIZE 13,136 COLOR 0,0,0,-1,-1,-1
@ PIXELS 310,162 GET Keywords->radio STYLE 65536 FONT "Times",268 PICTURE "@*C Radio" SIZE 13,55 COLOR 0,0,0,-1,-1,-1
@ PIXELS 40,333 GET Keywords->refugee co STYLE 65536 FONT "Times",268 PICTURE "@*C Refugee Control" SIZE 13,115 COLOR 0,0,0,-1,-1,-1
@ PIXELS 55,333 GET Keywords->r_and d STYLE 65536 FONT "Times",268 PICTURE "@*C Research and Development" SIZE 13,174 COLOR 0,0,0,-1,-1,-1
@ PIXELS 70,333 GET Keywords->safety STYLE 65536 FONT "Times",268 PICTURE "@*C Safety" SIZE 13,58 COLOR 0,0,0,-1,-1,-1
@ PIXELS 85,333 GET Keywords->satellites STYLE 65536 FONT "Times",268 PICTURE "@*C Satellites" SIZE 13,74 COLOR 0,0,0,-1,-1,-1
@ PIXELS 100,333 GET Keywords->sar STYLE 65536 FONT "Times",268 PICTURE "@*C Search and Rescue" SIZE 13,128 COLOR 0,0,0,-1,-1,-1
@ PIXELS 115,333 GET Keywords->security STYLE 65536 FONT "Times",268 PICTURE "@*C Security" SIZE 13,69 COLOR 0,0,0,-1,-1,-1
@ PIXELS 130,333 GET Keywords->state comp STYLE 65536 FONT "Times",268 PICTURE "@*C State Comprehensive" SIZE 13,141 COLOR 0,0,0,-1,-1,-1
@ PIXELS 145,333 GET Keywords->strikes STYLE 65536 FONT "Times",268 PICTURE "@*C Strikes" SIZE 13,62 COLOR 0,0,0,-1,-1,-1
@ PIXELS 160,333 GET Keywords->testing STYLE 65536 FONT "Times",268 PICTURE "@*C Testing" SIZE 13,64 COLOR 0,0,0,-1,-1,-1
@ PIXELS 175,333 GET Keywords->trade STYLE 65536 FONT "Times",268 PICTURE "@*C Trade" SIZE 13,54 COLOR 0,0,0,-1,-1,-1
@ PIXELS 190,333 GET Keywords->training STYLE 65536 FONT "Times",268 PICTURE "@*C Training" SIZE 13,70 COLOR 0,0,0,-1,-1,-1
@ PIXELS 205,333 GET Keywords->transporta STYLE 65536 FONT "Times",268 PICTURE "@*C Transportation" SIZE 13,105 COLOR 0,0,0,-1,-1,-1
@ PIXELS 220,333 GET Keywords->vessel_acc STYLE 65536 FONT "Times",268 PICTURE "@*C Vessel Accident" SIZE 13,112 COLOR 0,0,0,-1,-1,-1
@ PIXELS 235,333 GET Keywords->volcano STYLE 65536 FONT "Times",268 PICTURE "@*C Volcano" SIZE 13,68 COLOR 0,0,0,-1,-1,-1
@ PIXELS 250,333 GET Keywords->water_cont STYLE 65536 FONT "Times",268 PICTURE "@*C Water Control" SIZE 13,102 COLOR 0,0,0,-1,-1,-1
@ PIXELS 265,333 GET Keywords->water_poll STYLE 65536 FONT "Times",268 PICTURE "@*C Water Pollution" SIZE 13,111 COLOR 0,0,0,-1,-1,-1
@ PIXELS 280,333 GET Keywords->water qual STYLE 65536 FONT "Times",268 PICTURE "@*C Water Quality" SIZE 13,100 COLOR 0,0,0,-1,-1,-1
@ PIXELS 295,333 GET Keywords->weather STYLE 65536 FONT "Times",268 PICTURE "@*C Weather" SIZE 13,68 COLOR 0,0,0,-1,-1,-1
```

RETURN

\* EOF: GETs2.prg

\* Program...: Initialize

\*

\* ---Initialization of arrays "state" and "prov"

\*

\*

SELECT A

state (1)=Tba->ak

state (2)=Tba->wa

state (3)=Tba->id

state (4)=Tba->mt

state (5)=Tba->nd

state (6)=Tba->mn

state (7)=Tba->wi

state (8)=Tba->mi

state (9)=Tba->oh

state (10)=Tba->ny

state (11)=Tba->vt

state (12)=Tba->nh

state (13)=Tba->me

state (14)=Tba->or

state (15)=Tba->ca

prov (1)=Tba->yt

prov (2)=Tba->bc

prov (3)=Tba->ab

prov (4)=Tba->sk

prov (5)=Tba->mb

prov (6)=Tba->on

prov (7)=Tba->pq

prov (8)=Tba->nb

RETURN

\*

\* EOF : Initialize.prg

```

* Program...: MenuProc
*-----
* ---Menu Processing Program
*-----
* ---Name the values passed from MENU(0), MENU(1)
PARAMETERS the_menu,the_item
* ---Process input
DO CASE

    * ---If the selection was made in the position menu
    CASE the_menu = 1
        DO CASE

            * ---If the first option was selected "Top of File"
            CASE the_item = 1
                * - Go to the top of file
                GOTO TOP
                RETURN

            * ---If the second option was selected "Bottom of File"
            CASE the_item = 2
                * - Go to the bottom of file
                GOTO BOTTOM
                RETURN

            * ---The third option was a divider line

            * ---If the fourth option was selected "Next Record"
            CASE the_item = 4
                * ---Move forward one record
                SKIP
                * ---If end-of-file encountered
                IF EOF()
                    * ---Error message
                    ALERT NOTE 2 "You are positioned at the last record in the database."
                    * ---Reset record pointer to last record
                    GOTO BOTTOM
                ENDIF
                RETURN

            * ---If the fifth option was selected "Previous Record"
            CASE the_item = 5
                * - Go backward
                SKIP -1
                * ---If beginning-of-file encountered
                IF BOF()
                    * ---Error message
                    ALERT NOTE 2 "You are positioned at the first record in the database."
                    * ---Reset record pointer to first record
                    GOTO TOP
                ENDIF
                RETURN
        ENDCASE

    * ---If the selection was made in the record menu

```

```

CASE the_menu = 2
DO CASE

* ---If the first option was selected "Append a Record"
CASE the_item = 1
* ---Append a record
DO Add_Rec
RETURN

* ---If the second option was selected "Edit Record"
CASE the_item = 2
* ---Edit the current record
DO Edit_Rec
RETURN

* ---If the third option was selected "Delete Record"
CASE the_item = 3
* ---Delete the current record
SELECT A
STORE RECNO() TO delrec
SKIP -1
STORE RECNO() TO lastrec
GOTO delrec
DELETE
PACK
SELECT B
GOTO delrec
DELETE
PACK
SELECT A
GOTO lastrec
RETURN

* ---The fourth option was a divider line

* ---If the fifth option was selected "Locate a Record"
CASE the_item = 5
* ---Locate a record using the Find_Rec routine
DO Find_Rec
RETURN

* ---If the sixth option was selected "Continue"
CASE the_item = 6
* - Store the current record number to the lastrec variable
*   If record can not be LOCATED, program will return to this
*   record.
STORE RECNO() TO lastrec
* - Continue LOCATE
CONTINUE
* - If matching record was not found
IF .NOT. FOUND()
* - Display alert message
ALERT STOP 2 "No more matching records!"
* - Return to previous record location
GOTO lastrec

```

```

        * - Do the menu update (disable CONTINUE option)
        DO UpdtMenu
    ENDIF
    RETURN
ENDCASE

* ---If the selection was made in the utility menu
CASE the_menu = 3

    STORE RECNO() TO lastrec

    DO CASE

        * ---If the first option was selected "Report"
        CASE the_item = 1
            SET CONSOLE OFF
            REPORT FORM repcaus;
                FOR RECNO() = lastrec TO PRINT PROMPT
            SET CONSOLE ON
            GOTO lastrec
            RETURN

        * ---The second option was a divider line

        * ---If the third option was selected "Quit Application"
        CASE the_item = 3
            * - Exit application
            * ---
            CLOSE DATABASES
            SCREEN 1 OFF
            SCREEN 2 OFF
            * ---Indicate that processing is done
            STORE .T. TO done
            * ---Return to calling procedure'
            RETURN
    ENDCASE

ENDCASE

* EOF : MenuProc.prg

```

\* Program...: SAYS

\*-----

\* ---Display Program

\*-----

\*

SCREEN 1 TYPE 1 HEADING "Page 1" AT 29,9 SIZE 442,622 PIXELS FONT "Times",14 COLOR 0,0,0,22065,-1,21803  
@ PIXELS 10,4 SAY "Title" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 4,366 TO 65,616 STYLE 3871 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 21,375 SAY "Agreement ID" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 38,375 SAY "Canadian Reference Block" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 56,376 SAY "US Reference Block" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 69,4 SAY "Content" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 70,438 TO 132,616 STYLE 3871 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 85,446 SAY "Initial Date" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 103,446 SAY "Revision Date" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 123,447 SAY "Expiry Date" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 158,4 SAY "Canadian Authority" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 174,37 SAY "US Authority" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 197,4 SAY "States involved" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 200,5 TO 436,141 STYLE 3871 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 151,244 SAY "Emergency Type Addressed" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 171,251 SAY "Response Type Addressed" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 202,193 SAY "US Involvement" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 226,159 SAY "Canadian Involvement" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 250,173 SAY "Scale of Emergency" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 273,175 SAY "Authorization is for" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 301,150 SAY "Provinces/Territories involved" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 304,150 TO 435,286 STYLE 3871 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 197,383 SAY "Impact on other Agencies" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 274,383 SAY "References" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1  
@ PIXELS 349,383 SAY "Comments" STYLE 65536 FONT "Times",268 COLOR 0,0,0,-1,-1,-1

\*

RETURN

\*

\* EOF : SAYS.prg

\* Program...: SAYs2

\*-----

\* ---Display Program (show keywords)

\*-----

\*

SCREEN 2 TYPE 1 HEADING "Page 2" AT 124,103 SIZE 333,523 PIXELS FONT "Times",14 COLOR 0,0,0,-2120,-1,23624  
@ PIXELS 20,12 SAY "Keywords" STYLE 65536 FONT "Times",4366 COLOR 0,0,0,-1,-1,-1

\*

RETURN

\*

\* EOF : SAYs2.prg

```

*-----
*
* Program...: TBAtool.prg
* Date.....: February 1993
* By.....: Michèle Provencher
*
*-----
* --- SET Environment
*-----
*
SET SAFETY OFF           && overwrite files w/out confirmation
SET TALK OFF             && suppress interactive talk
SET BELL OFF             && bell will not sound on full field
ON ERROR DO Err_Proc    && do this when an error occurs
CLOSE ALL                && close all data files
*
*-----
* --- Open Files
*-----
*
USE "TBA.dbf"
SET INDEX TO "agr.idx"
SELECT B
USE "keywords.dbf"
SET INDEX TO "agr2.idx"
SELECT A
SET RELATION TO Tba->agr_id INTO KEYWORDS ADDITIVE
*
*-----
* --- RESTORE Memory Variables
*-----
*
RESTORE FROM tool ADDITIVE
STORE .F. to done
*
*-----
* --- Menu System
*-----
MENU BAR "Position;Record;Utilities"
*
* ---Define Position menu options
* ---Note /N & /P keyboard shortcuts and divider line
* ---Feel free to add /T & /B shortcuts to Top and Bottom options
MENU 1,"Top of File;Bottom of File;(-;Next/N;Previous/P"
*
* ---Define Record menu options
* ---Note /A, /E, /D, /L, /C keyboard shortcuts and divider line
MENU 2,"Append Record/A;Edit Record/E;Delete Record/D;(-;Locate/L;(Continue/C"
*
* --- Define Utilities menu options

```



```

* --- Note /R & /Q keyboard shortcuts and divider line
MENU 3,"Report/R;(-;Quit Application/Q"
*
* ---The following command activates the entire menu system.
*   When the user makes a selection from any one of the user-
*   defined menus, the DO MenuProc command will be executed.
*   When MenuProc is called, MENU(0) and MENU(1) pass the numbers
*   of the menu and option that was selected.
*
ON MENU DO MenuProc WITH MENU(0),MENU(1)
*
*-----
* --- Main
*-----
*
* ---We will then stay within the following loop until done is changed to True.
*   This change will only happen when you Quit the application.
PUBLIC temp1,temp2,temp3,temp4
PUBLIC state(15),prov(8)
DO WHILE .NOT. done
    CLEAR
    DO SAYS
    @ PIXELS 21,459 GET Tba->Agr_id STYLE 0 FONT "Times",14 SIZE 15,52 COLOR 0,0,0,-1,-1,-1
    DO GETS
    CLEAR GETS
    DO GETS1
    READ
ENDDO
*
*-----
* --- Closing Operations
*-----
*
ON ERROR
SET BELL ON
SET TALK ON
SET SAFETY ON
* ---Return to calling program
CLOSE ALL
RETURN
* END : TBAtool.prg

```

\* Program...: Update

\*-----  
\* ---Program to update the content of checkboxes and fields  
\*-----

\*

```
SELECT A
REPLACE Tba->ak WITH state(1)
REPLACE Tba->wa WITH state(2)
REPLACE Tba->id WITH state(3)
REPLACE Tba->mt WITH state(4)
REPLACE Tba->nd WITH state(5)
REPLACE Tba->mn WITH state(6)
REPLACE Tba->wi WITH state(7)
REPLACE Tba->mi WITH state(8)
REPLACE Tba->oh WITH state(9)
REPLACE Tba->ny WITH state(10)
REPLACE Tba->vt WITH state(11)
REPLACE Tba->nh WITH state(12)
REPLACE Tba->me WITH state(13)
REPLACE Tba->or WITH state(14)
REPLACE Tba->ca WITH state(15)
REPLACE Tba->yt WITH prov(1)
REPLACE Tba->bc WITH prov(2)
REPLACE Tba->ab WITH prov(3)
REPLACE Tba->sk WITH prov(4)
REPLACE Tba->mb WITH prov(5)
REPLACE Tba->on WITH prov(6)
REPLACE Tba->pq WITH prov(7)
REPLACE Tba->nb WITH prov(8)
RETURN
```

\*

\* EOF : Update.prg

```
* Program...: UpdtMenu
* -----
* ---Program to update the menu system
* -----
* ---This routine is only called after a LOCATE or CONTINUE option has been
*   executed. It properly enables or disables the CONTINUE option, based
*   on the results of the previous LOCATE or CONTINUE.
* ---If a record was successfully found
IF FOUND()
  * ---Enable the CONTINUE option
  MENU 2,"Append Record/A;Edit Record/E;Delete Record/D;(-;Locate/L;Continue/C"
ELSE
  && a matching record was not found
  * ---Disable the CONTINUE option
  MENU 2,"Append Record/A;Edit Record/E;Delete Record/D;(-;Locate/L;(Continue/C"
ENDIF
RETURN
*
* EOF : UpdtMenu.prg
```



